#!/usr/bin/python
Import ('env')

sources = env.Glob('*.cpp') + env.Glob('Joystick/*.cpp')

incs = '. #/source/kernel/gen_system #/intern/string'
incs += ' #/source/gameengine/Expressions #/intern/moto/include'
incs += ' #/source/gameengine/Rasterizer #/source/gameengine/SceneGraph'

defs = []

if env['WITH_BF_SDL']:
    incs += ' ' + env['BF_SDL_INC']
else:
    defs.append('DISABLE_SDL')

if env['WITH_BF_PYTHON']:
    incs += ' ' + env['BF_PYTHON_INC']
    defs.append('WITH_PYTHON')

if env['OURPLATFORM'] in ('win32-vc', 'win64-vc'):
    if env['BF_DEBUG']:
        defs.append('_DEBUG')

if env['WITH_BF_CXX_GUARDEDALLOC']:
    defs.append('WITH_CXX_GUARDEDALLOC')
    incs += ' #/intern/guardedalloc'

env.BlenderLib ( 'ge_logic', sources, Split(incs), defs, libtype=['core','player'], priority=[330,65], cxx_compileflags=env['BGE_CXXFLAGS'])
